Math 4530

Wednesday April 11

Isothermal coordinates and conformal maps

We look at pictures of conformal mappings. Maple maps coordinate curves to coordinate curves, so if you start with a grid of (small) squares, the image map will also be carved up into approximate squares, when your mapping is a conformal mapping of part of the plane. The conformal factor tells you how much the squares expand or shrink.

1) Illustrate the fact that stereographic projection is conformal (and its inverse is isothermal/conformal). We computed the conformal factor rho with Maple, on Monday. For the inverse of stereographic projection, it was 2/(u^2+v^2+1). This equals 1-z, where [x,y,z] is the image point on the sphere. Thus stereographic projection has conformal factor which is the reciprocal, namely 1/(1-z). These numbers are the factors by which infinitesimal length is scaled.

```
> restart:
  with(plots):
  with(linalg):
Warning, the name changecoords has been redefined

Warning, the protected names norm and trace have been redefined and unprotected
```

```
> sphere:=plot3d([2*u/(u^2+v^2+1),2*v/(u^2+v^2+1),(u^2+v^2-1)/(u^2+v
  ^2+1)],
  u=-2..2,v=-2..2,grid=[40,40]):
  plane:=plot3d([u,v,0],u=-1..1,v=-1..1,grid=[20,20]):
  display({sphere,plane}, title = "Stereographic projection is
  isothermal");
```

Stereographic projection is isothermal



\>

2) Import procedures to help compute the Gauss map

```
> #dot product
  dp:=proc(X,Y)
  X[1]*Y[1]+X[2]*Y[2]+X[3]*Y[3];
  end:
> #2-norm
  nrm:=proc(X)
  sqrt(dp(X,X));
  end:
> #cross product:
  xp:=proc(X,Y)
  local a,b,c;
  a:=X[2]*Y[3]-X[3]*Y[2];
  b:=X[3]*Y[1]-X[1]*Y[3];
  c:=X[1]*Y[2]-X[2]*Y[1];
  [a,b,c];
```

```
end:
> #Derivative matrix for mapping X:
DXq:=proc(X)
local Xu,Xv;
Xu:=matrix(3,1,[diff(X[1],u),diff(X[2],u),diff(X[3],u)]);
Xv:=matrix(3,1,[diff(X[1],v),diff(X[2],v),diff(X[3],v)]);
simplify(augment(Xu,Xv));
end:
> #Matrix of first fundamental form:
gij:=proc(X)
local g11,g12,g22,Y;
Y:=evalm(DXq(X));
simplify(evalm(transpose(Y)&*Y));
end:
> #unit normal:
N:=proc(X)
local Y,Z,s;
Y:=DXq(X);
Z:=xp(col(Y,1),col(Y,2));
s:=nrm(Z);
simplify(evalm((1/s)*Z));
end:
```

3) Illustrate the minimal surface diagram in which we discuss the composition of isothermal parameterization of minimal surface, Gauss map, and stereographic projection; the fact that such composition is conformal.

```
> Cat:=(u,v)->[cos(v)*cosh(u),sin(v)*cosh(u),u]:
    #catenoid parameterization
St:=(x,y,z)->[x/(1-z),y/(1-z),0]:
    #stereographic projection
gij(Cat(u,v));
   #verify conformal factor
N(Cat(u,v));
   #unit normal map
```

$$\begin{bmatrix} \cosh(u)^2 & 0 \\ 0 & \cosh(u)^2 \end{bmatrix}$$

$$\left[-\frac{\operatorname{csgn}(\cosh(u)^2)\cos(v)}{\cosh(u)}, -\frac{\operatorname{csgn}(\cosh(u)^2)\sin(v)}{\cosh(u)}, \frac{\sinh(u)\operatorname{csgn}(\cosh(u)^2)}{\cosh(u)}\right]$$

```
> plot3d([u,v,0],u=-Pi/2..Pi/2,v=0..2*Pi, grid=[20,40],
title="Paramterization domain, with square grid");
   #open set U, with square grid
```




```
> plot3d(Cat(u,v),u=-Pi/2..Pi/2,v=0..2*Pi,grid=[20,40],
  title="Catenoid, with isothermal parameterization");
    #catenoid piece, square grid

>
```




```
> plot3d(N(Cat(u,v)),u=-Pi/2..Pi/2,v=0..2*Pi,grid=[20,40],
  title="image of Gauss map, for Catenoid");
```




```
> plot3d(St(N(Cat(u,v))[1],N(Cat(u,v))[2],N(Cat(u,v))[3]),
```

```
  u=-Pi/2..Pi/2, v=0..2*Pi,grid=[20,40],
  title="triple composition (St)o(N)oX := g is conformal!");
```




```
> simplify(St(N(Cat(u,v))[1],N(Cat(u,v))[2],N(Cat(u,v))[3]));
```

$$\left[-\frac{\mathrm{csgn}(\cosh(u)^2)\cos(v)}{\cosh(u)-\sinh(u)\,\mathrm{csgn}(\cosh(u)^2)},\ -\frac{\mathrm{csgn}(\cosh(u)^2)\sin(v)}{\cosh(u)-\sinh(u)\,\mathrm{csgn}(\cosh(u)^2)},\ 0\right]$$

4) Complex analytic maps are conformal. In fact, conformal maps from the plane to the plane, when written in complex form f(z) either satisfy f(z) is analytic, or f(z bar) is analytic. More precisely, analytic maps are conformal maps which are orientation preserving. Here are 3 interesting examples:

```
> plot3d([u^2-v^2,2*u*v,0],u=0..1,v=0..1,
  title="f(z)=z^2");
>
```




```
> plot3d([r^2*cos(2*theta),r^2*sin(2*theta),0],r=0..Pi/2,
  theta=0..Pi/2,title="f(z)=z^2 in polar coordinates");
```




```
> plot3d([u/(u^2+v^2),-v/(u^2+v^2),0],u=(0.2)..2,v=(.2)..2,
```

```
  grid=[40,40],
  title="f(z)=1/z");
>
```




```
> plot3d([exp(u)*cos(v),exp(u)*sin(v),0],u=-Pi/2..Pi/2,
  v=0..2*Pi,grid=[20,40],title="f(z)=exp(z) - does this look
  familiar?");
>
```


4
2
0
–2
–4
–4
–2
0
2
4

>